IMPLEMENTATION OF CONCENTRATION DEPENDENT 'FIRST-PASS' MODELS USING NONMEM.

*L Phillips, **SR Cox.

*Pharmaceutical Outcomes Research, Inc. Williamsville, NY; **Pharmacia & Upjohn Kalamazoo, MI

ABSTRACT

The development of drugs extensively metabolized by the P450 enzyme system may require the need to model concentration dependent 'first-pass' effects. This simulation study, performed using NONMEM, presents two types of concentration dependent 'first-pass' models that might be used for drugs extensively metabolized by P450 enzymes located in the intestine. As shown in Figure 1, both models assume a non-linear concentration dependent 'first-pass' effect coupled with linear absorption and a combination of Michaelis-Menten (M-M) and linear elimination. The first model incorporates the 'first-pass' effect as a loss from the dose compartment and is implemented using equations specified in the $DES block in the control stream. The second model assumes the 'first-pass' effect occurs instantaneously by making bioavailability a non-linear function of concentration. This model is implemented in NONMEM with verbatim code in the $ERROR block of the control stream and by modifying the dosing record structure of the database. The behavior of the two models was simulated over several days of dosing using a range of values for Ka, Ke, and Vm, Km (M-M parameters). The simulations showed the most notable difference in the behavior of the two models to be in the relative approach of Cmax and Cmin to steady-state. The control streams for model implementation in NONMEM will be presented.

INTRODUCTION

The development of drugs extensively metabolized by the P450 enzyme system may require the need to model non-linear concentration dependent processes. Depending on the class of P450 enzymes and their location, non-linear concentration dependent processes may need to be incorporated as part of the pre-absorption, absorption, disposition, and/or elimination model. For example, drugs extensively metabolized by P450 enzymes located in the liver and intestine may require a non-linear concentration dependent 'first-pass' effect in addition to Michaelis-Menten elimination. Figure 1 shows two types of models that might be used to empirically describe the behavior of the above system. The first model incorporates the 'first-pass' effect as a non-linear loss from the depot (or dose) compartment. The second model assumes that the 'first-pass' effect occurs instantaneously by making bioavailability a non-linear function of concentration.

OBJECTIVE

The objective of the current simulation study is two-fold:

(1) to explore the implementation of the two models within NONMEM; and

(2) to explore differences in the behavior of the concentration-time profiles for the two models.

FIGURE 1

MODEL 1



MODEL 2



METHODS

The following steps were followed for each model:

(1) NONMEM Control Stream was generated.

(2) NONMEM (Version IV) was used to simulate concentration data using the following set of conditions:
Dose Groups: 150, 300, 600, 1200 mg
administered in three equal doses
Duration of Dosing: 8 days
Sampling Times: Every 15 minutes for 4 hours, followed by every 30 minutes until 8 hours after each dose

Pharmacokinetic Parameters

Parameter Description	Parameter (units)	Values
Absorption Rate Constant	Ka (1/hr)	1.5 or 3.0
Maximum Rate of Metabolism	Vm (mg/hr)	50 or 100
Amount to reach 50% of Vm	Km (mg)	50 or 100
Hepatic blood flow	Q (L/hr)	90
Volume of central compartment	Vc (L)	50
Elimination rate from Vc	(1/hr)	0.01

(3) SAS (Version 6.07) was used to calculate the following after each dose:
Cmin, Cmax, Tmax, Cdif = Cmax-Cmin

(4) To compare the behavior of the concentration-time profiles of the two models, the following plots were compared for each dose group and each set of pharmacokinetic parameters.
- Cmin vs. Time Since First Dose
- Cmax vs. Time Since First Dose
- Tmax vs. Time Since First Dose
- Cdif vs. Time Since First Dose

(5) Plots of the full sampling profile for the first and last dose of each model were also compared.

CONTROL STREAMS FOR NONMEM VERSION (IV)

MODEL 1

(1) Can be implemented with ADVANS 6, 8, or 9

(2) Requires $SUBROUTINE TOL, $MODEL, and $DES

```
$PROB - MODEL 1
$INPUT ID TIME AMT DV CMT EVID MDV
$DATA /data/model.nmdat
$SUBROUTINES ADVAN6 TRANS1 TOL=5
$MODEL
 COMP=(DEPOT,DEFDOSE,INITIALOFF)
 COMP=(CENTRAL,NODOSE,DEFOBS)
$PK
  KA=THETA(1)*(1+ETA(1))
  KM=THETA(2)*(1+ETA(2))
  VM=THETA(3)
  V= THETA(4)*(1+ETA(3))
  KE=THETA(5)
  S2=V/1000
$DES
  DEN1=KM+A(1)
  FR1=VM/DEN1
  DEN2=KM+A(2)
  FR2=VM/DEN2
  DADT(1)=-(KA+FR1)*A(1)
  DADT(2)=KA*A(1) - (FR2+KE)*A(2)
$ERROR
  Y = F * (1 + EPS(1))

 $THETA (0 , 1.5) (0 , 50) (0 , 100) (5 , 50)
(0 , 0.01)
 $OMEGA 0.3 0.3 0.3
 $SIGMA 0.3
 $ESTIMATION MAXEVAL=5000 PRINT=10
 $COV
 $TABLE ID TIME CMT  NOPRINT FILE= ../
model1.tbl NOHEADER
```

MODEL 2

(1) Requires a modified database structure
- NONMEM generally introduces dose to the PK system using the following DATA items:
  AMT - dose amount
  CMT - dose compartment number
  EVID - indicator variable describing type of observation
- Bioavailability is typically defined in the $PK block.
- Bioavailability in this model is dependent upon the current amount in the central compartment, therefore must be defined in the $ERROR block.
- The default dosing mechanisms can not apply an $ERROR block defined bioavailability.

(2) Can be implemented using ADVANS 6, 8, or 9

(3) Requires $SUBROUTINE TOL, $MODEL, $DES, and verbatim code

(4) Modified Database Structure

ID	TIME	DOSE	DV	CMT	EVID	MDV	TYPE
11	0	50	.	1	2	1	1
11	0.75	.	307.5	2	0	0	0
11	8	.	.	2	2	1	2
11	8	50	.	1	2	1	1
11	8.25	.	189.6	2	0	0	0

DOSE- dose amount (AMT can not be used)
EVID- =2 specifies an "other" event
=0 specifies a concentration event
TYPE- =1 specifies a dosing event
=2 specifies an update of the PK system to obtain the predicted current amount in the central compartment
=0 specifies a concentration event

```
$PROB - MODEL 2
$INPUT ID TIME DOSE DV CMT EVID MDV TYPE
$DATA /data/model2.nmdat
$SUBROUTINES ADVAN6 TRANS1 TOL=5
$MODEL
  COMP=(DEPOT,DEFDOSE,INITIALOFF)
  COMP=(CENTRAL,NODOSE,DEFOBS)
$PK
   KA=THETA(1)*(1+ETA(1))
   KM=THETA(2)*(1+ETA(2))
   VM=THETA(3)
   V=THETA(4)*(1+ETA(3))
   KE=THETA(5)*(1+ETA(4))
   Q=THETA(6)
   S2=V
$DES
   NUM=VM
   DEN=KM+A(2)
   DADT(1)=-KA*A(1)
   DADT(2)=KA*A(1) - (NUM/DEN+KE)*A(2)
```

```
$ERROR
1     R=(VM*V)/(KM+A(2))
2     BF=1-R/(R+Q)
3     A1=BF*DOSE+A(1)
4    " IF(EVID.EQ.2.AND.TYPE.EQ.1) THEN
5    " A(1)=A1
6    " DAETA(1,1)=D00085
7    " DAETA(1,2)=D00084
8    " DAETA(1,3)=D00083
9    " DAETA(1,4)=D00082
10   " ENDIF
11    Y=F * (1+EPS(1))

$THETA (0,1.5) (0,50) (0, 100) (5,50)
       (0, 0.01) (90 FIXED)
$OMEGA 0.3 0.3 0.3 0.3
$SIGMA 0.3
$EST MAXEVAL=5000 PRINT=10
$COV
$TABLE ID TIME DOSE CMT TYPE NOPRINT
       FILE= ../model2.tbl NOHEADER
```

(5) $ERROR block

- Lines 1 and 2 calculate bioavailability
- Line 3 adds the dose to the amount remaining in the depot compartment
- Line 4 determines if the record is a dose event
  - If the record is a dose event, Line 5 transfers the value of A1 into the depot compartment A(1).
  - Lines 6-9 provide the partial derivatives of A(1) with respect to each η contained in $PK.
  - If the record is not a dose event, Lines 5 - 9 are skipped.

(6) Obtaining partial derivatives of A(1)

- Remove all DAETA (A,B) lines from the control stream.
- Process control stream using NM-TRAN only.
- Review the FSUBS output file for the variable names of the partial derivatives.

```
A1=BF*DOSE+A(1)
C       D00078 = DERIVATIVE OF A1 W.R.T. ETA(04)
 D00078=DOSE*D00063
C       D00079 = DERIVATIVE OF A1 W.R.T. ETA(03)
 D00079=DOSE*D00062
C       D00080 = DERIVATIVE OF A1 W.R.T. ETA(02)
 D00080=DOSE*D00061
C        D00081 = DERIVATIVE OF A1 W.R.T. ETA(01)
 D00081=DOSE*D00060
C        D00082 = DERIVATIVE OF A1 W.R.T. ETA(04)
  D00082=DAETA(01,04)+D00078
C        D00083 = DERIVATIVE OF A1 W.R.T. ETA(03)
  D00083=DAETA(01,03)+D00079
C        D00084 = DERIVATIVE OF A1 W.R.T. ETA(02)
 D00084=DAETA(01,02)+D00080
C        D00085 = DERIVATIVE OF A1 W.R.T. ETA(01)
 D00085=DAETA(01,01)+D00081
```

- Add the DAETA (A, B) lines to the control stream.

GENERAL MODEL BEHAVIORS



STEADY-STATE Cmin (Cminss)

- The proportional change in Cminss values increased with dose and ranged from 2-79%.
- The proportional change in Cminss for the 600 and 1200 mg doses was 2-13 times larger than the proportional change in Cminss for the 300 and 600 mg dose groups.



STEADY-STATE Cmax (Cmaxss)

- The proportional change in Cmaxss values was fairly constant across dose (2.1 - 2.4%) except when Vm=50mg/hr (2.1 - 5.1%).
- Vm = 50 mg/hr and Doses = 600 mg and 1200 mg
  - Proportional changes in Cmaxss were higher than for other doses.
  - Proportional change in Cmaxss for Model 1 < Model 2.



STEADY-STATE Cdif (Cdifss)

- Model 1
  - Cdif was larger for first dose and declined to the Cdifss value.
- Model 2
  - For doses < 300 mg, Cdif was slightly larger for the first dose and declined to the Cdifss value.
  - For doses > 600 mg, Cdif was usually smaller after the first dose and increased to the Cdifss value.

DISCUSSION/CONCLUSION

CONTROL STREAMS

- NONMEM Version IV allows for the implementation of a variety of nonlinear PK models.
- Models with concentration dependent bioavailability require a special dataset structure.
- NONMEM steady-state and additional dosing structures are not available for use with Model 2.

MODEL BEHAVIOR

- Nonlinearity of the two systems is much more evident in Cminss vs. Dose than Cmaxss vs. Dose.
- The change in the peak-trough difference (Cdif) from first dose to steady-state demonstrates the most apparent difference between the two models.

MODEL SELECTION

- Drug specific simulations of a variety of models will elucidate the differences in model behaviors for varying parameter values and dosing regimens.
- Simulation results can be an important mechanism for model selection during data analysis.

ACKNOWLEDGEMENTS

The authors would like to gratefully acknowledge Dr. Stuart Beal for his help with coding the control stream for Model 2.

REFERENCES

NONMEM Users Guides, 1992, Beal, SL and Sheiner, LB (Eds.) NONMEM Project Group, Univeristy of California at San Francisco, San Francisco.